

Lesson 12: More Systems


```
[> restart;
```

▼ A geometry problem

Here's a nice little application of resultants to a geometrical problem. We're given two concentric circles with radii r_1 and r_2 . From a given point P at a distance d from the centre of the circles, we want to draw a line intersecting the circles at points Q_1 and Q_2 so that the distances PQ_2 and PQ_1 have a given ratio w . How should we do it?

Here's a picture in the case where $r_1 = 1$, $r_2 = 3$, $d = 2$. I'll try for $w = 2$ (but this line isn't the one that achieves that ratio).

```
> with(plots): with(plottools):
  display([
    circle([0,0],1),
    circle([0,0],3),
    plot([[2,0],[-2.840,.968]],colour=black),
    plot([[2,0],[0,0],[-.825,.565],[0,0],[-2.840,.968]],
  colour=green),
    textplot([[0,-.2,0],[2,0.2,P],[-0.9,0.8,Q[1]],
      [-2.9,1.3,Q[2]])]),
    axes=None, scaling=constrained);
```



A few things to notice about how I drew this picture:

- The circles were drawn with a command called **circle** in the **plottools** package. You tell it the coordinates of the centre of the circle (as a list $[x,y]$) and the radius. You could also give it a **colour** option. The default is black.
- The labels were printed with the **textplot** command, which is in the **plots** package. You give it a list consisting of x and y coordinates and the text you want it to print, or a list of such lists. I moved the coordinates slightly off the point I wanted to label, because I didn't want the label to be actually on the point. The text to print could be a string (enclosed in quotes), or a Maple expression.
- The **display** command is used to put all the pieces together in one plot.
- I gave the **display** command the option **axes = none**, because I didn't want axes interfering with the picture, and **scaling = constrained**, to make sure the circles look like circles.

Actually, let's write a procedure that will produce the plot for any Q_1 on the inner circle. We'll give it as input the angle $\theta = POQ_1$.

```
> drawpicture:= proc(theta)
    uses plots, plottools;
    local x1,y1,yline,x2,y2,x,r;
```

Press Ctrl-T to start a comment. It's ignored by Maple, but should help people understand the code.

To go back to Maple input, press Ctrl-M.

Another way to produce a comment, staying in the Maple input style, is to start the line with #

```
# The coordinates of Q1 are x1 and y1.
```

```
x1:= cos(theta);
```

```
y1:= sin(theta);
```

The equation of the line P Q1 in "point-slope" form is

$$(y-0)/(x-2) = \text{slope} = (y1-0)/(x1-2)$$

```
yline:= y1*(x-2)/(x1-2);
```

Q2 is the intersection of this line with the circle $x^2 + y^2 = 9$

Its coordinates are x2 and y2.

```
x2:= fsolve(x^2 + yline^2 = 9, x = -3 .. 0);
```

```
y2:= eval(yline, x = x2);
```

r is the ratio of distances P Q1 / P Q2

```
r := evalf(sqrt(((x2-2)^2 + y2^2)/((x1-2)^2 + y1^2)));
```

Finally, draw the picture.

```
display([
```

```
circle([0,0],1),
```

```
circle([0,0],3),
```

```
plot([[2,0],[x2,y2]],colour=black),
```

```
plot([[2,0],[0,0],[x1,y1],[0,0],[x2,y2]], colour=green),
```

```
textplot([[0,-.2,0],[2,0.2,P],[x1,y1+0.2,Q[1]],
```

```
[x2,y2+0.2,Q[2]]]),
```

```
axes=none, scaling=constrained, title=(ratio=r));
```

```
end proc;
```

```
drawpicture := proc(theta)
```

```
local x1, y1, yline, x2, y2, x, r;
```

```
x1 := cos(theta);
```

```
y1 := sin(theta);
```

```
yline := y1 * (x - 2) / (x1 - 2);
```

```
x2 := fsolve(x^2 + yline^2 = 9, x = -3 .. 0);
```

```
y2 := eval(yline, x = x2);
```

```
r := evalf(sqrt(((x2 - 2)^2 + y2^2) / ((x1 - 2)^2 + y1^2)));
```

```
plots:-display([plottools:-circle([0,0],1), plottools:-circle([0,0],3), plot([[2,0],
```

```
[x2,y2]], colour = black), plot([[2,0], [0,0], [x1,y1], [0,0], [x2,y2]], colour
```

```
= green), plots:-textplot([[0, -0.2, 0], [2, 0.2, P], [x1, y1 + 0.2, Q[1]], [x2, y2
```

```
+ 0.2, Q[2]])], axes = none, scaling = constrained, title = (ratio = r))
```


```
end proc
```

Note that the comments are not part of the Maple output from the procedure definition.

```
> drawpicture(3*Pi/4);
```

(1.1)

$ratio = 1.748501716$




Here's an animation.

```
> animate(drawpicture,[theta],theta=0..Pi);
```

ratio = 5.000000000

$\theta = 0.$



Now to translate the geometry into algebra.

Let α be the angle OPQ_1 , let $s_1 = PQ_1$ and $s_2 = PQ_2$. Then by the Law of Cosines,

$r_1^2 = d^2 + s_1^2 - 2 d s_1 \cos(\alpha)$. Similarly $r_2^2 = d^2 + s_2^2 - 2 d s_2 \cos(\alpha)$. We want $s_2 = s_1 w$. Now the first two equations involve a trig function (so not a polynomial), but they are polynomial in terms of $\cos(\alpha)$. So we define $a = \cos(\alpha)$.

```
> P[1]:= r[1]^2 - (d^2 + s[1]^2 - 2*d*s[1]*a);
```

```
P[2]:= r[2]^2 - (d^2 + (s[1]*w)^2 - 2*d*s[1]*w*a);
```

$$P_1 := r_1^2 - d^2 - s_1^2 + 2 d s_1 a$$

$$P_2 := r_2^2 - d^2 - s_1^2 w^2 + 2 d s_1 w a \quad (1.2)$$

These are polynomials in s_1 and a . They should both be 0 at the same s_1 when a is the cosine of the angle we're looking for.

```
> resultant(P[1],P[2],s[1]);
```

$$\begin{aligned} & w^4 r_1^4 - 2 w^4 r_1^2 d^2 - 2 w^2 r_1^2 r_2^2 + 2 w^2 r_1^2 d^2 + w^4 d^4 + 2 w^2 r_2^2 d^2 - 2 w^2 d^4 + r_2^4 - 2 r_2^2 d^2 + d^4 \\ & + 4 d^2 w^3 a^2 r_1^2 - 4 d^4 w^3 a^2 + 4 d^2 w a^2 r_2^2 - 4 d^4 w a^2 - 4 d^2 w^2 a^2 r_1^2 + 8 d^4 w^2 a^2 \\ & - 4 d^2 w^2 a^2 r_2^2 \end{aligned} \quad (1.3)$$

This should be 0.

```
> s:=solve(%,a);
```

$$S := \frac{1}{2} \frac{r_2^2 - d^2 + w^2 d^2 - w^2 r_1^2}{\sqrt{d^2 w^3 + d^2 w - 2 w^2 d^2 - w^3 r_1^2 - w r_2^2 + w^2 r_1^2 + w^2 r_2^2} d}, \quad (1.4)$$
$$- \frac{1}{2} \frac{r_2^2 - d^2 + w^2 d^2 - w^2 r_1^2}{\sqrt{d^2 w^3 + d^2 w - 2 w^2 d^2 - w^3 r_1^2 - w r_2^2 + w^2 r_1^2 + w^2 r_2^2} d}$$

It looks like two solutions, but not really: $-a$ and $-s_1$ give the same solution as a and s_1 . On the other hand, the angles α and $-\alpha$ both have the same cosine, which means you can reflect the picture across the x axis. Here is a in the case $r_1 = 1, r_2 = 3, d = 2, w = 2$.


```
> a1:= eval(S[1],{r[1]=1,r[2]=3,d=2,w=2});
```

$$a1 := \frac{17}{88} \sqrt{22} \quad (1.5)$$

```
> alpha1:= arccos(a1);
```

$$\alpha1 := \arccos\left(\frac{17}{88} \sqrt{22}\right) \quad (1.6)$$

To draw the picture we need the angle $\theta = POQ_1$. Some trig (the law of sines) relates α to θ .



```
> thetaeq:= sin(alpha1)/1 = sin(Pi-alpha1 - theta1)/2;
```


$$thetaeq := \frac{3}{88} \sqrt{154} = \frac{1}{2} \sin\left(\arccos\left(\frac{17}{88} \sqrt{22}\right) + \theta1\right)$$

(1.7)

```
> t1:= solve(thetaeq,theta1);
```

$$t1 := -\arccos\left(\frac{17}{88} \sqrt{22}\right) + \arcsin\left(\frac{3}{44} \sqrt{154}\right)$$

```
> drawpicture(t1);
```



Oops: wrong solution.

```
> solve(thetaeq, theta1, AllSolutions);
```

$$-\arccos\left(\frac{17}{88}\sqrt{22}\right) + \arcsin\left(\frac{3}{44}\sqrt{154}\right) - 2\arcsin\left(\frac{3}{44}\sqrt{154}\right)_{-B1} + 2\pi_{-Z1} + \pi_{-B1} \quad (1.8)$$

Of course a multiple of 2π won't make a difference.


```
> eval(%, _Z1=0);
```

$$-\arccos\left(\frac{17}{88}\sqrt{22}\right) + \arcsin\left(\frac{3}{44}\sqrt{154}\right) - 2\arcsin\left(\frac{3}{44}\sqrt{154}\right)_{-B1} + \pi_{-B1} \quad (1.9)$$

```
> thetas := eval(%, _B1=0), eval(%, _B1=1);
```

$$\text{thetas} := -\arccos\left(\frac{17}{88}\sqrt{22}\right) + \arcsin\left(\frac{3}{44}\sqrt{154}\right), -\arccos\left(\frac{17}{88}\sqrt{22}\right) - \arcsin\left(\frac{3}{44}\sqrt{154}\right) + \pi \quad (1.10)$$

```
> drawpicture(thetas[2]);
```

Here's a more "algebraic" way of getting θ from α :

The line PQ_2 has equation

```
> eqline := y = tan(alpha)*(2-x);
```

$$eqline := y = \frac{3}{374} \sqrt{154} \sqrt{22} (2-x) \quad (1.11)$$

Where does this intersect the smaller circle?

```
> intersects1 := solve({eqline, x^2 + y^2 = 1});
```

$$intersects1 := \left\{ x = -\frac{1}{8}, y = \frac{3}{176} \sqrt{154} \sqrt{22} \right\}, \left\{ x = \frac{37}{44}, y = \frac{9}{968} \sqrt{154} \sqrt{22} \right\} \quad (1.12)$$

Two solutions, of course. One is right. Where does it intersect the larger circle?

```
> intersects2 := solve({eqline, x^2 + y^2 = 9});
```

$$intersects2 := \left\{ x = -\frac{9}{4}, y = \frac{3}{88} \sqrt{154} \sqrt{22} \right\}, \left\{ x = \frac{261}{88}, y = -\frac{15}{1936} \sqrt{154} \sqrt{22} \right\} \quad (1.13)$$

What are the distances to the point P?

```
> distances1 := seq(eval(sqrt((2-x)^2 + y^2)), intersects1[i]), i
= 1..2);
```

$$\text{distances1} := \frac{1}{2} \sqrt{11} \sqrt{2}, \frac{1}{11} \sqrt{18} \sqrt{11}$$

```
> distances2 := seq(eval(sqrt((2-x)^2 + y^2), intersects2[i]), i
= 1..2);
```

$$\text{distances2} := \sqrt{22}, \frac{1}{22} \sqrt{25} \sqrt{22}$$

One of the `distances2` must be 2 times one of the `distances1`. We really don't need Maple to tell us which, but let's see if it can do that. The command `is` can be used (sometimes) to tell if an equation is true. I'll put an `if` statement inside two `for` loops to test all the possible pairs.

```
> for i from 1 to 2 do
  for j from 1 to 2 do
    if is(distances2[j] = 2*distances1[i]) then
      print(i,j)
    end if
  end do
end do:
```

1, 1 (1.14)

There's a slightly subtle point here. It wouldn't have worked if my conditional expression was just `distances2[j] = 2*distances1[i]`. This is because `distances2[1]` and `2*distances1[1]` are not literally the same, they are just mathematically equivalent.

```
> distances2[1] = 2*distances1[1];
```

$$\sqrt{22} = \sqrt{11} \sqrt{2} \quad (1.15)$$

```
> if distances2[1] = 2*distances2[1]
  then yes
  else no
end if;
```

no (1.16)

`if A = B ...` would just test whether A and B are literally the same. The `is` command tries harder to test whether the two things really are equal. Even it is not perfect: testing whether two expressions are equivalent can be a difficult task.


Anyway, we want the first point in `intersects1` and the first in `intersects2`.

```
> theta := eval(arctan(y/x), intersects1[1]);
```

$$\theta := -\arctan\left(\frac{3}{22} \sqrt{154} \sqrt{22}\right) \quad (1.17)$$

```
> drawpicture(theta);
```

ratio = 2.177042958




Oops: still not right. The wrong angle with that tan. There is a two-parameter form of **arctan** that is better: $\text{arctan}(y, x)$ gives the angle from the positive x axis to the point $[x, y]$.

```
> theta := eval(arctan(y, x), intersects1[1]);
```

$$\theta := -\arctan\left(\frac{3}{22} \sqrt{154} \sqrt{22}\right) + \pi$$

(1.18)

```
> drawpicture(theta);
```



Three equations in three variables

After such success with two polynomials in two variables, what about more polynomials in more variables, say three?

Consider polynomials $p_1(x, y, z), p_2(x, y, z), p_3(x, y, z)$. We might try something like the following:

- $p_4(x, y) = \text{resultant}(p_1(x, y, z), p_2(x, y, z), z)$. This is 0 for any x, y such that, for some z , both $p_1(x, y, z)$ and $p_2(x, y, z)$ are 0.
- $p_5(x, y) = \text{resultant}(p_1(x, y, z), p_3(x, y, z), z)$. This is 0 for any x, y such that, for some z , both $p_1(x, y, z)$ and $p_3(x, y, z)$ are 0.
- Then solve the two-variable system $\{p_4(x, y) = 0, p_5(x, y) = 0\}$ as before.
- Finally, find z by using those values of x and y in the equations $p_1(x, y, z) = 0, p_2(x, y, z) = 0, p_3(x, y, z) = 0$.

Unfortunately this doesn't always work.

```
> p1 := 1-z-y-y*z-y^2+3*x*z-x*y+7*x^2;
```

```

p2 := 2-z+z^2+x*z-x*y;
p3 := -1+2*z^2-y-y*z+y^2-x*z;

```

$$\begin{aligned}
p1 &:= 1 - z - y - yz - y^2 + 3xz - xy + 7x^2 \\
p2 &:= 2 - z + z^2 + xz - xy \\
p3 &:= -1 + 2z^2 - y - yz + y^2 - xz
\end{aligned}$$

```

> p4 := resultant(p1, p2, z);
p5 := resultant(p1, p3, z);

```

$$p4 := 2 - 8x + 22x^2 + 3y^2 - 17xy - 16yx^2 - 4xy^2 - 5x^2y^2 - 13x^3y + 3y^3 + 28x^4 + y^4 + 28x^3 + 2y$$

$$p5 := 1 + 5x + 22x^2 - 4y^2 + 11xy - 46yx^2 + 4xy^2 - 29x^2y^2 - 17x^3y - 3y^3x + 7y^3 + 119x^4 + 4y^4 - 7x^3 - 8y$$

```

> s := solve({p4,p5},{x,y});

```

$$\begin{aligned}
S := & \left\{ x = \text{RootOf}(1337 _Z^8 - 63 _Z^7 + 10193 _Z^6 - 9114 _Z^5 - 2382 _Z^4 - 11866 _Z^3 \right. \\
& + 7488 _Z^2 + 2524 _Z + 2332), y = \frac{15143581232100125}{5988643946241287} \text{RootOf}(1337 _Z^8 - 63 _Z^7 \\
& + 10193 _Z^6 - 9114 _Z^5 - 2382 _Z^4 - 11866 _Z^3 + 7488 _Z^2 + 2524 _Z + 2332) \\
& - \frac{4087892364931924}{5988643946241287} + \frac{1113748060342526}{5988643946241287} \text{RootOf}(1337 _Z^8 - 63 _Z^7 \\
& + 10193 _Z^6 - 9114 _Z^5 - 2382 _Z^4 - 11866 _Z^3 + 7488 _Z^2 + 2524 _Z + 2332)^7 \\
& - \frac{1605286237578695}{5988643946241287} \text{RootOf}(1337 _Z^8 - 63 _Z^7 + 10193 _Z^6 - 9114 _Z^5 \\
& - 2382 _Z^4 - 11866 _Z^3 + 7488 _Z^2 + 2524 _Z + 2332)^6 \\
& + \frac{16545376071926893}{11977287892482574} \text{RootOf}(1337 _Z^8 - 63 _Z^7 + 10193 _Z^6 - 9114 _Z^5 \\
& - 2382 _Z^4 - 11866 _Z^3 + 7488 _Z^2 + 2524 _Z + 2332)^5 \\
& - \frac{36323228548883043}{11977287892482574} \text{RootOf}(1337 _Z^8 - 63 _Z^7 + 10193 _Z^6 - 9114 _Z^5 \\
& - 2382 _Z^4 - 11866 _Z^3 + 7488 _Z^2 + 2524 _Z + 2332)^4 \\
& + \frac{30619107969747}{22727301503762} \text{RootOf}(1337 _Z^8 - 63 _Z^7 + 10193 _Z^6 - 9114 _Z^5 \\
& - 2382 _Z^4 - 11866 _Z^3 + 7488 _Z^2 + 2524 _Z + 2332)^3 \\
& + \frac{189205432961553}{352273173308311} \text{RootOf}(1337 _Z^8 - 63 _Z^7 + 10193 _Z^6 - 9114 _Z^5
\end{aligned}$$

$$\left. \begin{aligned} & -2382 _Z^4 - 11866 _Z^3 + 7488 _Z^2 + 2524 _Z + 2332 \Big)^2, \{x=1, y=2\}, \left\{x=-\frac{1}{5}, y \right. \\ & \left. = -\frac{8}{5} \right\} \end{aligned} \right\}$$

Ignoring the complicated solution using RootOf, notice the two nice solutions $\{x=1, y=2\}$ and $\left\{x=-\frac{1}{5}, y=-\frac{8}{5}\right\}$. Do these lead to solutions of the original system?

$$\begin{aligned} & > \text{eval}([p1, p2, p3], \{x=1, y=2\}); \\ & \qquad \qquad \qquad [0, z^2, 1 + 2z^2 - 3z] \end{aligned} \tag{2.1}$$

Clearly this won't work: z would have to be 0 for p_2 to be 0, but that won't work in p_3 .

$$\begin{aligned} & > \text{eval}([p1, p2, p3], \{x=-1/5, y=-8/5\}); \\ & \qquad \qquad \qquad \left[0, \frac{42}{25} - \frac{6}{5}z + z^2, \frac{79}{25} + 2z^2 + \frac{9}{5}z\right] \end{aligned} \tag{2.2}$$

We can use resultant to see if the last two have any roots in common.

$$\begin{aligned} & > \text{resultant}(\%[2], \%[3], z); \\ & \qquad \qquad \qquad \frac{17917}{625} \end{aligned} \tag{2.3}$$

No they don't. What went wrong?

$p_4(x, y) = 0$ for those x, y such that, for some z , both $p_1(x, y, z)$ and $p_2(x, y, z)$ are 0.

$p_5(x, y) = 0$ for those x, y such that, for some z , both $p_1(x, y, z)$ and $p_3(x, y, z)$ are 0.

The trouble is that the z that makes $p_1(x, y, z) = p_2(x, y, z) = 0$ might not be the same z that makes $p_1(x, y, z) = p_3(x, y, z) = 0$.

In this particular example, with $x=1$ and $y=2$,

for $z=0$ we have $p_1(1, 2, 0) = p_2(1, 2, 0) = 0$, but $p_3(1, 2, 0) \neq 0$,

while for $z=1$ we have $p_1(1, 2, 1) = p_3(1, 2, 1) = 0$ but $p_2(1, 2, 1) \neq 0$.

There are more advanced methods that can be used, e.g. Gröbner bases, but we won't be able to go into that. In any case, **solve** does know about those methods, and can be used.

$$\begin{aligned} & > \text{solve}(\{p1=0, p2=0, p3=0\}); \\ & \left\{x = \frac{3496971470650}{4480899278621} + \frac{98904476427751}{94098884851041} \text{RootOf}(2772 - 11592 _Z + 22640 _Z^2 \right. \\ & \qquad \qquad \qquad - 23002 _Z^3 + 16070 _Z^4 - 5940 _Z^5 + 603 _Z^6 + 437 _Z^7 + 191 _Z^8) \\ & \qquad \qquad \qquad - \frac{26054438772371}{13442697835863} \text{RootOf}(2772 - 11592 _Z + 22640 _Z^2 - 23002 _Z^3 \\ & \qquad \qquad \qquad + 16070 _Z^4 - 5940 _Z^5 + 603 _Z^6 + 437 _Z^7 + 191 _Z^8)^2 \\ & \qquad \qquad \qquad + \frac{227818874044747}{94098884851041} \text{RootOf}(2772 - 11592 _Z + 22640 _Z^2 - 23002 _Z^3 \\ & \qquad \qquad \qquad + 16070 _Z^4 - 5940 _Z^5 + 603 _Z^6 + 437 _Z^7 + 191 _Z^8)^3 \\ & \qquad \qquad \qquad \left. - \frac{93029075861483}{62732589900694} \text{RootOf}(2772 - 11592 _Z + 22640 _Z^2 - 23002 _Z^3 \right\} \end{aligned} \tag{2.4}$$

$$\begin{aligned}
& + 16070 _Z^4 - 5940 _Z^5 + 603 _Z^6 + 437 _Z^7 + 191 _Z^8)^4 \\
& + \frac{28598806512911}{62732589900694} \text{RootOf}(2772 - 11592 _Z + 22640 _Z^2 - 23002 _Z^3 \\
& + 16070 _Z^4 - 5940 _Z^5 + 603 _Z^6 + 437 _Z^7 + 191 _Z^8)^5 \\
& + \frac{42962168137871}{188197769702082} \text{RootOf}(2772 - 11592 _Z + 22640 _Z^2 - 23002 _Z^3 \\
& + 16070 _Z^4 - 5940 _Z^5 + 603 _Z^6 + 437 _Z^7 + 191 _Z^8)^6 \\
& + \frac{5725329607453}{94098884851041} \text{RootOf}(2772 - 11592 _Z + 22640 _Z^2 - 23002 _Z^3 \\
& + 16070 _Z^4 - 5940 _Z^5 + 603 _Z^6 + 437 _Z^7 + 191 _Z^8)^7, y = \frac{11264996410234}{4480899278621} \\
& - \frac{17590855192423}{13442697835863} \text{RootOf}(2772 - 11592 _Z + 22640 _Z^2 - 23002 _Z^3 + 16070 _Z^4 \\
& - 5940 _Z^5 + 603 _Z^6 + 437 _Z^7 + 191 _Z^8) + \frac{21156388925612}{13442697835863} \text{RootOf}(2772 \\
& - 11592 _Z + 22640 _Z^2 - 23002 _Z^3 + 16070 _Z^4 - 5940 _Z^5 + 603 _Z^6 + 437 _Z^7 \\
& + 191 _Z^8)^2 + \frac{30400243677697}{26885395671726} \text{RootOf}(2772 - 11592 _Z + 22640 _Z^2 \\
& - 23002 _Z^3 + 16070 _Z^4 - 5940 _Z^5 + 603 _Z^6 + 437 _Z^7 + 191 _Z^8)^3 \\
& - \frac{7764470791961}{8961798557242} \text{RootOf}(2772 - 11592 _Z + 22640 _Z^2 - 23002 _Z^3 \\
& + 16070 _Z^4 - 5940 _Z^5 + 603 _Z^6 + 437 _Z^7 + 191 _Z^8)^4 \\
& + \frac{2572252318973}{4480899278621} \text{RootOf}(2772 - 11592 _Z + 22640 _Z^2 - 23002 _Z^3 \\
& + 16070 _Z^4 - 5940 _Z^5 + 603 _Z^6 + 437 _Z^7 + 191 _Z^8)^5 \\
& + \frac{6661566329341}{26885395671726} \text{RootOf}(2772 - 11592 _Z + 22640 _Z^2 - 23002 _Z^3 \\
& + 16070 _Z^4 - 5940 _Z^5 + 603 _Z^6 + 437 _Z^7 + 191 _Z^8)^6 \\
& + \frac{1400872196737}{26885395671726} \text{RootOf}(2772 - 11592 _Z + 22640 _Z^2 - 23002 _Z^3 \\
& + 16070 _Z^4 - 5940 _Z^5 + 603 _Z^6 + 437 _Z^7 + 191 _Z^8)^7, z = \text{RootOf}(2772 \\
& - 11592 _Z + 22640 _Z^2 - 23002 _Z^3 + 16070 _Z^4 - 5940 _Z^5 + 603 _Z^6 + 437 _Z^7 \\
& + 191 _Z^8) \}
\end{aligned}$$

> evalf([allvalues(%)]);

```
[ {x = -0.4865575321 - 0.8484869687 I, y = -0.4978608316 + 1.646866383 I, z = 1.106698456 + 0.7706687592 I}, {x = 1.024255751 + 0.08128602695 I, y = 2.046083414 + 0.1702488504 I, z = 0.4522943811 + 0.3272142302 I}, {x = -0.2020169332 + 0.4105433001 I, y = -1.124809592 + 0.7785966800 I, z = 0.2847216570 + 1.163442961 I}, {x = -0.312121 - 2.855112 I, y = 4.0347020 + 5.5031388 I, z = -2.987693552 + 2.986942207 I}, {x = -0.312121 + 2.855112 I, y = 4.0347020 - 5.5031388 I, z = -2.987693552 - 2.986942207 I}, {x = -0.2020169332 - 0.4105433001 I, y = -1.124809592 - 0.7785966800 I, z = 0.2847216570 - 1.163442961 I}, {x = 1.024255751 - 0.08128602695 I, y = 2.046083414 - 0.1702488504 I, z = 0.4522943811 - 0.3272142302 I}, {x = -0.4865575321 + 0.8484869687 I, y = -0.4978608316 - 1.646866383 I, z = 1.106698456 - 0.7706687592 I}]
```

```
> remove(has, %, I);
```

```
[ ]
```

It turns out this system of equations has no real solutions, just complex ones.

▼ Vectors and Matrices

In preparation for introducing Newton's method for systems of equations in several variables, I want to show how Maple deals with vectors and matrices. Here's a 3-component column vector in Maple:

```
> C := < a, b, c >;
```

$$C := \begin{bmatrix} a \\ b \\ c \end{bmatrix}$$

(3.1)

Actually, I should say that's a **Vector**. There are two separate data structures in Maple, **vector** and **Vector**, and similarly there are both **matrix** and **Matrix**. The lower-case **vector** and **matrix** structures are from an old package called **linalg**, which is pretty much obsolete, but has been kept around for the sake of backwards compatibility (i.e. people still have programs that were written in older versions of Maple, and want them to still work). We'll only use the new-style structures.

Here's a row Vector. In output it looks rather like a list, except that the entries are separated by spaces instead of commas.

```
> R := < a | b | c >;
```

$$R := [a \ b \ c]$$

(3.2)

So within the "<" and ">", "," is used to separate items vertically and "|" to separate them horizontally. Here's a 3 x 3 Matrix. You can think of it as three column Vectors side by side.

```
> M := <<1,2,3>|<4,5,6>|<7,8,10>>;
```

$$M := \begin{bmatrix} 1 & 4 & 7 \\ 2 & 5 & 8 \\ 3 & 6 & 10 \end{bmatrix}$$

(3.3)

The same Matrix could have been entered by rows instead of by columns:

```
> <<1|4|7>, <2|5|8>, <3|6|10>>;
```

$$\begin{bmatrix} 1 & 4 & 7 \\ 2 & 5 & 8 \\ 3 & 6 & 10 \end{bmatrix}$$

(3.4)

You can add or subtract Vectors and Matrices of the same shape using + or -, as you might expect.

```
> <<a,b>|<c,d>> + <<1|2>, <3|4>>;
```

$$\begin{bmatrix} a+1 & c+2 \\ b+3 & d+4 \end{bmatrix}$$

(3.5)

```
> <a,b> + <c|d>;
```

Error, (in rtable/Sum) invalid arguments

I did say "the same shape". You can't add a column Vector and a row Vector, they have different shapes. You could convert from one to the other: this operation is called "transpose", and can be done with ^%T:

```
> <a,b>^%T; <a,b>^%T + <c|d>;
```

$$\begin{bmatrix} a & b \\ a+c & b+d \end{bmatrix}$$

(3.6)

You can multiply or divide Matrices and Vectors by scalars using * or /, as you might expect.

```
> 3*<a|b> + <c|d>/2;
```

$$\begin{bmatrix} 3a + \frac{1}{2}c & 3b + \frac{1}{2}d \end{bmatrix}$$

(3.7)

But the multiplication sign for Matrices and Vectors is . instead of *. Of course, what you multiply should be compatible: number of columns of the Matrix or Vector on the left must be equal to the number of rows of the one on the right.

```
> R . C;
```

$$a^2 + b^2 + c^2$$

(3.8)

```
> C . R;
```

$$\begin{bmatrix} a^2 & ab & ac \\ ab & b^2 & bc \\ ac & bc & c^2 \end{bmatrix}$$

(3.9)

```
> M . R;
```

Error, (in LinearAlgebra:-Multiply) cannot multiply a Matrix and row Vector

```
> R . M;
```

$$\begin{bmatrix} a+2b+3c & 4a+5b+6c & 7a+8b+10c \end{bmatrix}$$

(3.10)

```
> M . C;
```

$$\begin{bmatrix} a + 4b + 7c \\ 2a + 5b + 8c \\ 3a + 6b + 10c \end{bmatrix} \quad (3.11)$$

You can take an integer power of a square Matrix using \wedge .

```
> M^2;
```

$$\begin{bmatrix} 30 & 66 & 109 \\ 36 & 81 & 134 \\ 45 & 102 & 169 \end{bmatrix} \quad (3.12)$$

That extends to negative powers too if the Matrix is invertible.

```
> M^(-1);
```

$$\begin{bmatrix} -\frac{2}{3} & -\frac{2}{3} & 1 \\ -\frac{4}{3} & \frac{11}{3} & -2 \\ 1 & -2 & 1 \end{bmatrix} \quad (3.13)$$

```
> % . M;
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad (3.14)$$

```
> <<1,1>|<1,1>>^(-1);
```

Error, (in rtable/Power) singular matrix

There are very big and capable **LinearAlgebra** and **VectorCalculus** packages that can do lots of things with Vectors and Matrices, but we won't need to go much beyond this very basic level.

I'll need one command from the **VectorCalculus** package: **Jacobian**.

Consider a column Vector V whose components are expressions depending on several variables.

```
> V := <f(x,y,z), g(x,y,z), h(x,y,z)>;
```

$$V := \begin{bmatrix} f(x, y, z) \\ g(x, y, z) \\ h(x, y, z) \end{bmatrix} \quad (3.15)$$

The **Jacobian** of F is the Matrix of partial derivatives of the components of F with respect to each of the variables. Since this is the only member of the **VectorCalculus** package I'll want to use today, and I want to avoid some side-effects of **VectorCalculus**, I'll only take this one procedure from **VectorCalculus**. You can do that with an extra input to **with**.

```
> with(VectorCalculus, Jacobian);
```

[Jacobian] (3.16)

```
> J := Jacobian(V, [x, y, z]);
```

$$J := \begin{bmatrix} \frac{\partial}{\partial x} f(x, y, z) & \frac{\partial}{\partial y} f(x, y, z) & \frac{\partial}{\partial z} f(x, y, z) \\ \frac{\partial}{\partial x} g(x, y, z) & \frac{\partial}{\partial y} g(x, y, z) & \frac{\partial}{\partial z} g(x, y, z) \\ \frac{\partial}{\partial x} h(x, y, z) & \frac{\partial}{\partial y} h(x, y, z) & \frac{\partial}{\partial z} h(x, y, z) \end{bmatrix} \quad (3.17)$$

Notice that the rows of J correspond to the components of V , and the columns correspond to the variables. You could think of each row of J as the gradient vector of a component of V .

It will be convenient for us to consider functions from Vectors to Vectors. You can define these like this:

$$\begin{aligned} > F := X \rightarrow \langle X[1] + X[2]*X[3], X[2] - X[1]^2, \cos(X[3]) \rangle; \\ & F := X \rightarrow \langle X_1 + X_2 X_3, X_2 - X_1^2, \cos(X_3) \rangle \end{aligned} \quad (3.18)$$

> F(<a,b,c>);

$$\begin{bmatrix} a + b c \\ b - a^2 \\ \cos(c) \end{bmatrix} \quad (3.19)$$

Maple objects introduced in this lesson

is

arctan

Matrix

Vector

<...>, ,, | for constructing Vectors and Matrices

. for multiplying Vectors and Matrices

^%T

LinearAlgebra package

VectorCalculus package

Jacobian (in **VectorCalculus** package)